

## Using ODS TAGSETS.RTF to Fit Text and Multiple Graphs on One Page

Teresa Wilson, The Emmes Corporation

### ABSTRACT

A common challenge is to fit a lot of information on a single page of an RTF document. When the usual solutions with the Output Delivery System (ODS) RTF statement don't work, the ODS TAGSETS.RTF statement offers an alternative solution for managing space on the page efficiently. For example, it is possible to fit two graphs on one page, along with a title and subtitle for the page and captions for each graph. This can be done by using the options that come with ODS TAGSETS.RTF and by creating a style template with the TEMPLATE procedure to decrease the space between elements on the page.

### INTRODUCTION

You are creating an RTF document using ODS RTF, and you want to fit several elements like graphs and text on one page to save space. It looks like there is enough space, but you can't seem to place elements any closer together. You don't have control over the vertical spacing, or rather SAS<sup>®</sup> doesn't have the control.


However, if you use ODS TAGSETS.RTF, SAS gives you finer control over vertical spacing using the PARSKIP style element. This paper describes PARSKIP and other options that you can add to your toolbox to achieve optimal spacing and layout.

### ODS RTF VS. ODS TAGSETS.RTF

The Output Delivery System (ODS) has two destinations that produce Rich Text Format (RTF) output: ODS RTF and ODS TAGSETS.RTF. RTF documents are useful because you can create reports that you can edit further in Microsoft Word and other word processing programs. With ODS RTF you have only limited control over page layout, especially in the vertical direction. Instead, Microsoft Word controls things like page breaks and placement of titles and footnotes.

By default, ODS RTF does the following: It places continuous section breaks between PROCs, formats titles and footnotes as tables, and places titles and footnotes in the header and footer of the document (unless you use the BODYTITLE option in the ODS RTF statement). It also inserts blank lines between tables, and you don't have control over the height or number of these blank lines. You need to make changes to these features within a word processing program.

Figure 1 shows how output created with ODS RTF contains vertical space that cannot be changed using SAS. (Note: The shaded rectangles represent blank lines and are for illustration purposes only.) The output on the left is without the BODYTITLE option, and the output on the right is with BODYTITLE. The BODYTITLE option moves the titles and footnotes into the body of the document, but it also adds extra lines after the graphs and the output now spreads to a second page.



**Figure 1. Output from ODS RTF: Without (left) and with (right) BODYTITLE**


You can't control these extra blank lines with ODS RTF, but ODS TAGSETS.RTF does give you that option. TAGSETS.RTF is an ODS destination that was introduced in SAS 9.2. It gives the user more control over page breaks and space between text elements. In other words, it provides more control over space in the vertical direction.

TAGSETS.RTF uses RTF as its output type and is defined by a tagset. A tagset in SAS is "a template that defines how to create a type of markup language output from a SAS format" (SAS 9.4 Output Delivery System: User's Guide, p. 1191). TAGSETS.RTF provides specific style elements and options that control the layout beyond what traditional RTF can do.

## THE PARSKIP STYLE ELEMENT

The TAGSETS.RTF destination gives you access to the PARSKIP style element, which controls the space before and after tables. This style element is not available with ODS RTF. PARSKIP inherits its characteristics from the TitlesAndFooters style element, and you can use it to increase or decrease space around text.

Figure 2 shows the output produced with ODS TAGSETS.RTF. The yellow highlighting indicates the vertical space controlled by PARSKIP. This is the space that can be adjusted using SAS. Note that the additional blank lines shown in Figure 1 that were produced with ODS RTF are no longer present.



**Figure 2. Output from ODS TAGSETS.RTF with Default PARSKIP Size**

You can access the PARSKIP style element by using the TEMPLATE procedure to create a new style that specifies the size of the PARSKIP element:

```
proc template;
  define style mystyle;
    parent=styles.rtf;
    style parskip / fontsize=10pt;
  end;
run;
```

The PARENT statement defines a style based on the default STYLES.RTF. Any subsequent STYLE statements modify that default style.

When you specify PARSKIP, you also have to specify the FONTSIZE attribute, which is the height of the vertical space that SAS inserts. Note that this is the size of the blank line and not the font size of the text on the page. The default value of the PARSKIP FONTSIZE is 11pt. The smallest FONTSIZE that you can specify is 1 pt (1/72 inch). (But there is a way to get to zero pt; see the VSPACE suboption later in this paper.)

To illustrate how to use PARSKIP and other options to adjust the vertical space on the page, we will generate one RTF page with two graphs, along with titles, subtitles, and footnotes. (The complete program to create this output is in the Appendix at the end of this paper.) For this example, we create three styles: one for the main document, one for graph titles, and one for footnotes. Each of these three styles (“docstyle”, “grtitle”, and “ftnote”) has a different PARSKIP size (10pt, 5pt, and 1pt, respectively), which will let us look at the behavior of PARSKIP in detail as its value changes in a document.

## OPTIONS FOR MANAGING SPACE

Several options for the ODS TAGSETS.RTF statement affect vertical spacing:

```
%let pathout = C:\Users\twilson\Documents;
ods tagsets.rtf file="&pathout.\myreport.rtf"
               style=docstyle
               options (tables_off="systitleandfootercontainer ")
               startpage=no
               nogtitle
               nogfootnote;
```

### STYLE

The ODS TAGSETS.RTF statement is where you specify the particular style that sets the PARSKIP size. The style in the STYLE= option and the corresponding PARSKIP size stay in effect until a different style is specified.

### TABLES\_OFF

The OPTIONS option was introduced with ODS TAGSETS.RTF for options that are specific to TAGSETS.RTF. The TABLES\_OFF= suboption controls whether text is put into tables or into the body of the document.

By default, ODS TAGSETS.RTF places text like titles, footnotes, and TEXT= into tables. When you specify OPTIONS (TABLES\_OFF= "**style\_elements**"), SAS renders plain text instead of tables for those style elements. By turning off tables, you gain a little space vertically because plain text takes up less space than tables. It may also make subsequent editing of the document easier.

How do you know which style element to specify with TABLES\_OFF? Run your program with the suboption set to TABLES\_OFF="STYLE\_ELEMENTS". This writes the names of the style elements for your output in the log, and then you can find the name for the specific element you want to change.

### STARTPAGE

With TAGSETS.RTF, you can use many of the same options from traditional RTF. STARTPAGE=NO prevents a page break at the start of each procedure. (The default is YES.) This allows you to put multiple graphs on one page.

### NOGTITLE AND NOGFOOTNOTE

The options NOGTITLE and NOGFOOTNOTE place the titles and footnotes for the graph in the body of the RTF file instead of in the graph. This gives you control over the placement and spacing of the titles and footnotes.

### AXIS LENGTH=

When you generate a plot, you can specify the LENGTH option in the AXIS statement:

```
* x-axis;
axis1 label=(f=arial h=9pt) value=(f=arial h=8pt);

* y-axis;
axis2 length=75pct label=(angle=90 f=arial h=9pt) value=(f=arial h=8pt);

proc gplot data=sashelp.iris;
    plot SepalLength*SepalWidth=Species /
        noframe
        haxis=axis1 vaxis=axis2;
run;
quit;
```

The LENGTH option in the AXIS statement for the y-axis shrinks the graph area in the vertical direction, which adds more white space above and below the graph. In this example, the adjustment is in units of percentage of output area.

## ADDING TEXT

You can add text to a document in various ways.

### TITLE

You can use TITLE statements for the page title and the first graph title. Blank titles can add additional space:

```
title1 h=12pt bold font=arial "Fitting Text and Multiple Graphs on One
Page";
title2 h=12pt bold font=arial "Using Options Available with ODS
TAGSETS.RTF";
title3 h=12pt bold font=arial "Including PARSKIP Style Element";
title4 h=12pt " ";
title5 h=10pt bold font=arial "Title for Graph #1 Created with TITLE
Statement";
```

### ODS TAGSETS.RTF TEXT=

You can use the ODS TAGSETS.RTF TEXT= statement for graph titles, footnotes, and any other text on page. The TEXT= option allows you to insert text on the page and specify the PARSKIP spacing with a defined style ("grtitle", in this case):

```
ods tagsets.rtf
text="Title for Graph #2 Created with ODS TAGSETS.RTF TEXT="
style=grtitle
options(tables_off="usertext");
```

Why do you need to use ODS TAGSETS.RTF TEXT= rather than ODS TEXT=? You could use ODS TEXT=, but the PARSKIP style element is only available with ODS TAGSETS.RTF.

### Specify new style

Set the style for the graph title with the STYLE= option to change the value of the PARSKIP spacing.

### TABLES\_OFF= suboption


For the graph title, you need to specify the style element USERTEXT in the TABLES\_OFF suboption because that is the element created by TEXT=.

### FOOTNOTES

For the footnotes we use a new style, "fnote", to change the PARSKIP spacing to a size of 1pt:

```
ods tagsets.rtf
text="The footnote style sets the size of the PARSKIP element."
style=fnote;
```

Figure 3 shows the final output created using these options. The complete program is in the Appendix.



**Figure 3. Final Output for Program in Appendix**

The 10pt PARSKIP takes effect when the “docstyle” style is specified in the first ODS TAGSETS.RTF statement, which opens the document. PARSKIP changes to 5pt when the “grtitle” style is specified at the title for Graph #2. The PARSKIP in the “fnote” style has no visible effect because no output follows the footnotes. The LENGTH option for AXIS adds empty space directly above and below each graph.

## FURTHER ADJUSTMENTS

Several other options are available if you need to fine-tune the spacing on the page.

## MARGINS

You can change the margins by creating a new style and specifying any or all of BOTTOMMARGIN, TOPMARGIN, LEFTMARGIN, and RIGHTMARGIN. This new style creates narrow margins:

```
proc template;
  define style docstyle_narrow;
    parent=styles.rtf;
    style parskip / fontsize=10pt;
    style body from body/
      bottommargin=0.25in
      topmargin=0.25in
      leftmargin=0.25in
      rightmargin=0.25in;
  end;
run;
```

## FONT SIZE

You can also increase or decrease the text FONTSIZE and change the FONTFAMILY to adjust the space:

```
proc template;
  define style ftnote;
    parent=styles.rtf;
    style parskip / fontsize=1pt;
    style usertext from Note/
      just=center
      textalign=left
      fontweight=medium
      fontsize=10pt
      fontfamily=arial;
  end;
run;
```

## VSIZE, HSIZE

The options VSIZE and HSIZE in the GOPTIONS statement adjust the vertical and horizontal graph size in units of IN or CM (inches or centimeters):

```
goptions vsize=3in hsize=4.5in;
```


## VSPACE

VSPACE is a new suboption to OPTIONS that was introduced in the first maintenance release of SAS 9.4 (TS1M1). You can specify OPTIONS (VSPACE="NO") to remove all space before and after tables:

```
ods tagsets.rtf options(vspace="no");
```

The VSPACE suboption overrides the PARSKIP style element, so you can use one or the other but not both in the same section of code. However, you can switch between them on the same page. By setting VSPACE="YES", you can turn PARSKIP back on again.

Use VSPACE="NO" when you need to place elements as close together as possible. Figure 4 shows the effect of the VSPACE suboption. Compare this to Figure 2, and you can see that all PARSKIP elements have been removed.



**Figure 4. The Effect of the VSPACE Suboption**

## CONCLUSION

You have many options for document layout and vertical spacing when using ODS TAGSETS.RTF. The right combination of options can produce a well-formatted page with the desired spacing. The additional level of control over vertical spacing is one reason to choose ODS TAGSETS.RTF over ODS RTF.

## APPENDIX

The complete program to print two graphs and multiple lines of text on one page is below:

```
ods html close;

* Style for document;
proc template;
  define style docstyle;
    parent=styles.rtf;
    style parskip / fontsize=10pt;
    style body from body /
      bottommargin=1in
      topmargin=1in
      leftmargin=1.75in
      rightmargin=1.75in;
  end;
run;
```

```

* Style for graph titles;
proc template;
  define style grtitle;
    parent=styles.rtf;
    style parskip / fontsize=5pt;
    style usertext from Note /
      just=center
      textalign=center
      fontweight=bold
      fontsize=10pt
      fontfamily=arial;
  end;
run;

* Style for footnotes;
proc template;
  define style ftnote;
    parent=styles.rtf;
    style parskip / fontsize=1pt;
    style usertext from Note /
      just=center
      textalign=left
      fontweight=medium
      fontsize=10pt
      fontfamily=arial;
  end;
run;

* Set up output;
%let pathout=C:\Users\twilson\Documents;

options orientation='portrait'
  nodate
  nonumber;

goptions reset=all
  device=png300
  vsize=3in
  hsize=4.5in;

ods tagsets.rtf file="&pathout.\myreport.rtf"
  style=docstyle
  options (tables_off="systitleandfootercontainer")
  startpage=no
  nogtitle
  nogfootnote;

title;
footnote;

* Page Titles;
title1 h=14pt bold font=arial "Fitting Text and Multiple Graphs on One
Page";
title2 h=14pt bold font=arial "Using ODS TAGSETS.RTF";
title3 h=12pt bold font=arial "The PARSKIP Style Element in Action";
title4 h=12pt " ";

```

```

* Title for Graph #1;
title5 h=10pt bold font=arial "Title for Graph #1 Created with TITLE
Statement";

* x-axis;
axis1 label=(f=arial h=9pt) value=(f=arial h=8pt);

* y-axis;
axis2 length=75pct label=(angle=90 f=arial h=9pt) value=(f=arial h=8pt);

* Data point symbols;
symbol1 value=plus color=black height=2.7;
symbol2 value=triangle color=black height=2.7;
symbol3 value=circle color=black height=2.7;

* Legend;
legend1 across=1 down=3 noframe
      position=(bottom right inside)
      label=(position=top f="arial" h=2.7)
      value=(f="arial" h=2.7)
      shape=symbol(5,2)
      mode=protect;

* Graph #1;
proc gplot data=sashelp.iris;
  plot SepalLength*SepalWidth=Species /
    noframe
    haxis=axis1 vaxis=axis2
    legend=legend1;
run;
quit;

* Title for Graph #2;
ods tagsets.rtf
  text="Title for Graph #2 Created with ODS TAGSETS.RTF TEXT="
  style=grtitle
  options(tables_off="usertext");

* Graph #2;
proc gplot data=sashelp.iris;
  plot PetalLength*PetalWidth=Species /
    noframe
    haxis=axis1 vaxis=axis2
    legend=legend1;
run;
quit;

* Footnotes;
ods tagsets.rtf
  text="The footnote style sets a new PARSKIP size but has no visible
effect."
  style=ftnote;
ods tagsets.rtf
  text="The same style also decreases the font size of the text."
  style=ftnote;

```

```
ods tagsets.rtf close;  
ods html;
```

## REFERENCES

Hester, Wayne. 2006. "Teaching Your RTF Tagset to Do Clever Tricks." *Proceedings of the SAS Users Group International 31 2006 Conference*, Cary, NC: SAS Institute Inc. Available at <http://www2.sas.com/proceedings/sugi31/067-31.pdf>.

Meng, Xiangxiang. 2010. "ODS TAGSETS.RTF: Tips and Tricks." *Proceedings of the MidWest SAS Users Group 2010 Conference*, Cary, NC: SAS Institute Inc.. Available at <https://www.mwsug.org/proceedings/2010/dataviz/MWSUG-2010-67.pdf>.

SAS® 9.4 Output Delivery System: User's Guide, Fifth Edition.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Teresa Wilson  
The Emmes Corporation  
[twilson@emmes.com](mailto:twilson@emmes.com)